REDOR

tested 190817 using *SpinDynamica* 3.0.1 under *Mathematica* 11.0

init

```
Needs["SpinDynamica`"]
```

```
SpinDynamica version 3.0.1 loaded
```

ModifyBuiltIn: The following built–in routines have been modified in SpinDynamica:
{Chop, Dot, Duration, Exp, Expand, ExpandAll, NumericQ, Plus, Power, Simplify, Times, WignerD}.
Evaluate ??*symbol* to generate the additional definitions for *symbol*.

code

spin system & Hamiltonians

define heteronuclear 2-spin-1/2 system

```
SetSpinSystem[{{"I", 1/2}, {"S", 1/2}}]
```

SetSpinSystem: the spin system has been set to $\{\{I, \frac{1}{2}\}, \{S, \frac{1}{2}\}\}$

SetBasis: the state basis has been set to ZeemanBasis$[\{\{I, \frac{1}{2}\}, \{S, \frac{1}{2}\}\}$, BasisLabels → Automatic$]$.

heteronuclear 2-spin-1/2 Hamiltonian in a rotating solid

```
ωIS[t_, ΩMR_] := {0, 0, bIS, 0, 0}.WignerD[2, {{0}}][{ΩPMIS, ΩMR, {αRL0 - ωr t, βRL, 0}}]
```

```
HIS[ΩMR_] := PeriodicFunction[t,
   Evaluate[(2 π / ωr)],
   Evaluate@Chop@N[ωIS[t, ΩMR] 2 opI["I", "z"].opI["S", "z"]]
  ]
```

example

```
ΩPMIS = {0, 0, 0};
ΩMR = {0, π/2, 0};
αRL0 = 0;
HIS[ΩMR]
```

WignerD: The built-in function WignerD has been given extra functionality in SpinDynamica. Execute ?WignerD for more information.

For ?WignerD click <here>

$$\text{PeriodicFunction}\Big[t, \frac{2\pi}{\omega r},\; 2.\,\text{bIS}\,(I_z{\bullet}S_z)\,\Big(0.5\,\big(0.5 - 1.5\,\text{Cos}[\beta\text{RL}]^2\big) + 0.612372\,\big(0.306186\times 2.71828^{(0.-2.\,i)\,t\,\omega r} - 0.306186\times 2.71828^{(0.-2.\,i)\,t\,\omega r}\,\text{Cos}[0.5\,\beta\text{RL}]^4 + 1.83712\times 2.71828^{(0.-2.\,i)\,t\,\omega r}\,\text{Cos}[0.5\,\beta\text{RL}]^2\,\text{Sin}[0.5\,\beta\text{RL}]^2 - 0.306186\times 2.71828^{(0.-2.\,i)\,t\,\omega r}\,\text{Sin}[0.5\,\beta\text{RL}]^4\big) + 0.612372\,\big(0.306186\times 2.71828^{(0.+2.\,i)\,t\,\omega r} - 0.306186\times 2.71828^{(0.+2.\,i)\,t\,\omega r}\,\text{Cos}[0.5\,\beta\text{RL}]^4 + 1.83712\times 2.71828^{(0.+2.\,i)\,t\,\omega r}\,\text{Cos}[0.5\,\beta\text{RL}]^2\,\text{Sin}[0.5\,\beta\text{RL}]^2 - 0.306186\times 2.71828^{(0.+2.\,i)\,t\,\omega r}\,\text{Sin}[0.5\,\beta\text{RL}]^4\big)\Big)\Big]$$

default parameters

default parameters (5 kHz spinning frequency, 200Hz dipolar coupling)

```
ωr = 2 π 5 × 10^3;
βRL = ArcTan[Sqrt[2]];
bIS = 2 π 200;
ΩPMIS = {0, 0, 0};
ΩMR = {0, π / 4, 0};
αRL0 = 0;
τr = 2 π / ωr
```

$$\frac{1}{5000}$$

define REDOR sequence

```
REDORsequence[n_] :=
  {
   Repeat[
    {
     {None, τr / 2},
     RotationSuperoperator["S", {π, 0}],
     {None, τr / 2},
     RotationSuperoperator["S", {π, 0}]
    }, n
   ],
   RotationSuperoperator["I", {π, 0}],
   Repeat[
    {RotationSuperoperator["S", {π, 0}],
     {None, τr / 2},
     RotationSuperoperator["I", {π, 0}],
     {None, τr / 2}
    }, n
   ],
   RotationSuperoperator["I", {π / 2, 0}]
  };
```

calculations

calculate trajectories of in-phase and antiphase magnetization, during the REDOR sequence, for a single molecular orientation

```
ΩMR = {0, π / 3, π / 3};
```

trajectories of I-spin x-magnetization (blue) and antiphase 2IySz terms (red) during the sequence

```
{Ixtraj, IySztraj} = Trajectory[
  opI["I", "x"] -> {opI["I", "x"], 2 opI["I", "y"].opI["S", "z"]},
  REDORsequence[10],
  BackgroundGenerator → HIS[ΩMR]
 ]
```

```
{TrajectoryFunction[ {{0, 4. × 10^-3}} , <>], TrajectoryFunction[ {{0, 4. × 10^-3}} , <>]}
```

```
Plot[{Re@Ixtraj[tms × 10^-3], Re@IySztraj[tms × 10^-3]},
 {tms, 0, 10^3 × EventDuration[REDORsequence[10]]},
 Frame → True, PlotStyle → {{Thick, Blue}, {Thick, Red}},
 LabelStyle → Directive[Medium, FontFamily → "Helvetica"],
 FrameLabel → {"time /ms", "amplitude"}
]
```



note how the antiphase terms (red) build up while the in-phase terms (blue) decay

single orientation

```
ListPlot[
 TransformationAmplitudeTable[
  opI["I", "x"] -> opI["I", "x"],
  REDORsequence[n],
  {n, 0, 50, 2},
  BackgroundGenerator → HIS[ΩMR],
  TableCoordinates → EventDuration[REDORsequence[n]] × 10^3
 ],
 Joined → True, Frame → True, PlotStyle → Thick, PlotRange → {-1, 1},
 FrameLabel → {"pulse sequence duration / ms", None},
 LabelStyle → Directive[Medium, FontFamily → "Helvetica"]
]
```

Get: Cannot open CloudObjectLoader`.

SetOperatorBasis: the operator basis has been set to

ShiftAndZOperatorBasis[{{I, $\frac{1}{2}$}, {S, $\frac{1}{2}$}}, Sorted → CoherenceOrder].



calculate powder average trajectory of in-phase magnetization (REDOR curve)

powder average

```
ΩMR =.;
table =
  TransformationAmplitudeTable[
   opI["I", "x"] -> opI["I", "x"],
   REDORsequence[n],
   {n, 0, 50, 2},
   BackgroundGenerator → HIS[ΩMR],
   TableCoordinates → EventDuration[REDORsequence[n]] × 10^3,
   EnsembleAverage → {ΩMR, OrientationsAndWeights["ZCW50"]}
  ];
```

```
Predefined orientational sampling schemes:
 {Leboct10, Leboct16, Leboct19, Leboct22, Leboct31, Leboct37, Leboct46,
  Leboct85, POLYTOPE12, POLYTOPE60, Randomαβ, Randomαβγ, Randomβ, REPULSION100,
  REPULSION150, REPULSION168, REPULSION232, REPULSION376, REPULSION700,
  Stepαβ, Stepβ, ZCW1154, ZCW144, ZCW200, ZCW300, ZCW50, ZCW538, ZCW6044}
 Execute OrientationalSamplingScheme[scheme] for the usage message of a sampling scheme.
```

```
ListPlot[
 Re@table,
 Joined → True, Frame → True, PlotStyle → Thick, PlotRange → All, Axes → True,
 FrameLabel → {"pulse sequence duration / ms", None},
 LabelStyle → Directive[Medium, FontFamily → "Helvetica"]
]
```